********************************************************************************
* Do file for analysis (last updated on Feb. 27, 2023)
*
* TITLE  : Does Peacekeeping by Civilians Work? Reducing Armed Violence without Armed Force
* JOURNAL: Japanese Journal of Political Science
* AUTHOR : ASANO, Rui
*
* All data analyses in this file were carried out using Stata/MP 15.1 for Mac (64-bit Intel).
********************************************************************************

/*Setting*/
	cls
	version 15.1
	clear all
	*set more off, perm
	*set matsize 11000, perm
	*set scheme s1mono, perm
	*ssc install estout
	cd "[set your directory]"
	/*Please make a folder named "Figures" in the "Data" folder in case you save figures created in this do file.*/





********************************************************************************
/*Dataset description*/
	use "Data/02. Peacekeeping/UNPO6_financial_resources_base.dta", clear /*Note: Please check Dataset_monthly.dta or UNPO6_..._conflict-month.dta regarding the description of the variables on UN PKO financial resources.*/
	drop if MissionAbbrev == "UNOMSA" | MissionAbbrev == "UNAMIS" /*2 obs and 1 ob; these two missions are not PKOs nor SPMs.*/

	/*# of years*/
	gen year  = year(PeriodFrom)
	gen year2 = year(PeriodTo)
	sum year* /*1988-2019*/

	/*# of missions*/
	tab MissionAbbrev
	return list        /*56 missions*/
	tab Mission2Abbrev /*+2: UNCRO and UNMOP are not listed for MissionAbbrev.*/
	tab Mission3Abbrev /*+1: UNTMIH is not listed for MissionAbbrev.*/
	dis 56 + 2 + 1     /*59 UN PKOs included in the dataset*/

	/*# of documents*/
	tab DocumentSymbol
	return list       /*431 obs; 420 documents -> more than 400 tables (even if UNOMSA and UNAMIS are included)*/




********************************************************************************
/*Figures*/
/*Figure 1*/
	gen t = 1988 in 1
	replace t = t[_n-1] + 1 in 2/31 /*Note: t denotes the starting year of budgetary periods.*/

	gen UniformedExp = MilitaryExp + PoliceExp
	gen AllPKExp     = MilitaryExp + PoliceExp + CivilianExp
	gen AllExp       = MilitaryExp + PoliceExp + CivilianExp + OpsExp
	gen military     = .
	gen uniformed    = .
	gen allpk        = .
	gen all          = .

	gen num_missions = .
	foreach num of numlist 1988/2018 {
		quietly sum MilitaryExp if year == `num'
		replace military = r(sum)/1000 if t == `num'
		quietly sum UniformedExp if year == `num'
		replace uniformed = r(sum)/1000 if t == `num'
		quietly sum AllPKExp if year == `num'
		replace allpk = r(sum)/1000 if t == `num'
		quietly sum AllExp if year == `num'
		replace all = r(sum)/1000 if t == `num'
		quietly tab MissionAbbrev if year == `num'
		replace num_missions = r(r) if t == `num'
	}
	*br if Mission2Abbrev != "" | Mission3Abbrev != ""
	replace num_missions = num_missions + 1 if t >= 1996 & t <= 2002 /*UNMOP*/
	replace num_missions = num_missions + 1 if t == 1991             /*UNTAC*/
	replace num_missions = num_missions + 3 if t == 1997             /*UNPSG, UNSMIH, UNTMIH*/
	replace num_missions = num_missions + 1 if t == 1994             /*UNOMUR*/
	replace num_missions = num_missions + 1 if t >= 1999 & t <= 2000 /*UNOMSIL*/
	replace num_missions = num_missions + 1 if t >= 1995 & t <= 1996 /*UNCRO,  UNPREDEP*/

	twoway ///
		bar  military            t, lc(black) lw(vvthin) fc(blue%50)  || ///
		rbar uniformed military  t, lc(black) lw(vvthin) fc(green%50) || ///
		rbar allpk     uniformed t, lc(black) lw(vvthin) fc(pink%50)  || ///
		rbar all       allpk     t, lc(black) lw(vvthin) fc(gray%50)  || ///
		line num_missions        t, lc(black) lp(dash) yaxis(2)          ///
		ylabel(0(2000)8000, angle(0) format(%12.0fc) grid)      ///
		ylabel(0(5)20, angle(0) axis(2))                        ///
		ytitle("Sum of expenditures (thousand USD)" " ")        ///
		ytitle("# of missions", axis(2) orientation(rvertical)) ///
		xlabel(1988 2000 2010 2018) xtitle("")                  ///
		legend(order(1 "Military" 2 "Police" 3 "Civilian" 4 "Operations" 5 "# of missions") ///
		region(lc(none)) row(2)) plotregion(lc(none))
	*graph export "Data/Figures/Fig01_Sum_of_expenditures_over_time.pdf", as(pdf) replace

	list military        if t == 2000 | t == 2018
	dis 2643.885 / 846.7564                       /*3.12 times*/
	list uniformed allpk if t == 2000 | t == 2018
	dis (4478.801-3035.386) / (1636.26-1069.203)  /*2.55 times*/
	gen milprop = military / all
	gen civprop = (allpk - uniformed) / all
	twoway connect milprop t || connect civprop t, ylabel(,grid)


/*Figure 2*/
	tab CountryName
	tab MissionAbbrev
	tab Mission2Abbrev
	tab Mission3Abbrev

	gen location = CountryName
	/*Domestic areas*/
	replace location = TerritoryName if TerritoryName != ""
	/*Special cases for surface areas
	  Note1: UNOMUR was deployed to the Ugandan counties, but deployed for a conflict in Rwanda.
	         Because the financial resources for UNOMUR were reported with UNAMIR at some time,
			 UNOMUR is regarded as having been deployed to Rwanda in the figures.
	  Note2: UNMOP was deployed to a part of Croatia, but regarded as having been deployed to Bosnia and Herzegovina
	         because the financial resources for UNMOP were all reported with UNMIBH.
	  Note3: UNPROFOR was deployed even after the Yugoslavia was dissolved, but regarded
			 as having been deployed to (the former) Yugoslavia for all periods for convenience.*/
	replace location = "CAR, Chad"       if MissionAbbrev == "MINURCAT"
	replace location = "Croatia"         if MissionAbbrev == "UNTAES"   | MissionAbbrev == "UNPSG" /*Eastern Slavonia, Baranja, and Western Sirmium*/
	replace location = "Central America" if MissionAbbrev == "ONUCA"
	replace location = "Yugoslavia"      if MissionAbbrev == "UNPROFOR"
	/*Multiple missions at separate times*/
	replace location = "CAR (-2000)"         if MissionAbbrev == "MINURCA"
	replace location = "CAR (2014-)"         if MissionAbbrev == "MINUSCA"
	replace location = "Haiti (-2000)"       if MissionAbbrev == "UNMIH"    | MissionAbbrev == "UNSMIH"    | MissionAbbrev == "MIPONUH"
	replace location = "Haiti (2004-)"       if MissionAbbrev == "MINUSTAH" | MissionAbbrev == "MINUJUSTH"
	replace location = "Liberia (-1997)"     if MissionAbbrev == "UNOMIL"
	replace location = "Liberia (2003-)"     if MissionAbbrev == "UNMIL"
	replace location = "Timor-Leste (-2005)" if MissionAbbrev == "UNMISET"  | MissionAbbrev == "UNTAET"
	replace location = "Timor-Leste (2006-)" if MissionAbbrev == "UNMIT"
	/*Long names -> abbreviated*/
	replace location = "DRC"       if location == "Democratic Republic of the Congo"
	replace location = "Macedonia" if location == "The former Yugoslav Republic of Macedonia" /*Now, North Macedonia*/

	gen location2 = location if location == "Côte d'Ivoire" |  location == "Haiti (2004-)" | location == "Liberia (2003-)"
	twoway ///
		connected MilitaryExp PeriodFrom, lc(blue%50)  lw(thick) m(Oh) mc(blue)  || ///
		connected PoliceExp   PeriodFrom, lc(green%50) lw(thick) m(Dh) mc(green) || ///
		connected CivilianExp PeriodFrom, lc(pink%50)  lw(thick) m(Th) mc(pink)     ///
		by(location2, row(1) note("")) subtitle(, fc(none) lc(none))                ///
		xlabel(15706 17532 19724 21185, format(%tdCCYY) labsize(medsmall))          ///
		ylabel(0(50000)350000, grid gmax format(%9.0gc) angle(0) labsize(medsmall)) ///
		legend(order(1 "Military" 2 "Police" 3 "Civilian") row(1) region(lc(none) fc(none)) size(medsmall)) ///
		xtitle("") ytitle("Expenditures (thousand USD)") plotregion(lc(none))
	*graph export "Data/Figures/Fig02_Examples_expenditures_time_trends.pdf", as(pdf) replace


/*Figure 3*/
	sort location PeriodFrom
	by location: gen time = _n
	by location: egen sum_GrossExp = sum(GrossExp)
	foreach var of varlist MilitaryExp PoliceExp CivilianExp {
		by location: egen sum_`var' = sum(`var')
		gen pct_`var' = sum_`var' / sum_GrossExp
	}

	graph hbar (sum) pct_MilitaryExp pct_PoliceExp pct_CivilianExp if time == 1,     ///
		stack over(location, label(labsize(vsmall)) sort(3) descending axis(noline)) ///
		bar(1, fc(blue%50)  lc(black) lw(vvthin))  ///
		bar(2, fc(green%50) lc(black) lw(vvthin))  ///
		bar(3, fc(pink%50)  lc(black) lw(vvthin))  ///
		ylabel(,labsize(small) glw(vthin))         ///
		b1title("Ratio to gross total expenditures", size(small))                    ///
		legend(order(1 "Military" 2 "Police" 3 "Civilian") region(lc(none) fc(none)) ///
		symxsize(2) symysize(2) size(vsmall) keygap(1) ring(0) position(5) row(3))   ///
		plotregion(lc(none)) xsize(1) ysize(1.4)
	*graph export "Data/Figures/Fig03_Component_ratios_to_total_expenditures_by_location.pdf", as(pdf) replace


/*Figure 4*/
	keep if Region == "Africa"
	/*In the world map, it would be difficult to identify color darkness outside of Africa
	  because most of the host countries outside of Africa are small.*/

	gen NAME_0 = CountryName
	replace NAME_0 = TerritoryName if TerritoryName != ""
	replace NAME_0 = "Chad"      if MissionAbbrev == "MINURCAT"
	replace NAME_0 = "Eritrea"   if MissionAbbrev == "UNMEE"
	replace NAME_0 = "Sudan"     if MissionAbbrev == "UNISFA" | MissionAbbrev == "UNAMID"
	*tab NAME_0

	gen days = mdy(month(PeriodTo), day(PeriodTo), year(PeriodTo)) - mdy(month(PeriodFrom), day(PeriodFrom), year(PeriodFrom)) + 1
	*gen militaryexp_days = MilitaryExp / days
	*gen policeexp_days   = PoliceExp   / days
	gen civilianexp_days = CivilianExp / days
	bysort NAME_0: egen max_civilianexp_days = max(civilianexp_days)

	duplicates drop NAME_0, force
	keep NAME_0 max_civilianexp_days
	*save "Data/02. Peacekeeping/UNPO8_financial_resources_base_map.dta", replace

	cd "Data/04. Maps"
	*spshape2dta "../../RawData/04. Maps/gadm36_levels_shp/gadm36_0.shp", replace
	use "gadm36_0.dta", clear
	keep if ///
		NAME_0=="Algeria"       | NAME_0=="Angola"      | NAME_0=="Benin"         | NAME_0=="Botswana"      | ///
		NAME_0=="Burkina Faso"  | NAME_0=="Burundi"     | NAME_0=="Cameroon"      | NAME_0=="Cape Verde"    | ///
		NAME_0=="Central African Republic"              | NAME_0=="Chad"          | NAME_0=="Comoros"       | ///
		NAME_0=="Republic of Congo"                     | NAME_0=="Democratic Republic of the Congo"        | ///
		NAME_0=="Djibouti"      | NAME_0=="Egypt"       | NAME_0=="Equatorial Guinea"                       | ///
		NAME_0=="Eritrea"       | NAME_0=="Ethiopia"    | NAME_0=="Gabon"         | NAME_0=="Gambia"        | ///
		NAME_0=="Ghana"         | NAME_0=="Guinea"      | NAME_0=="Guinea-Bissau" | NAME_0=="Côte d'Ivoire" | ///
		NAME_0=="Kenya"         | NAME_0=="Lesotho"     | NAME_0=="Liberia"       | NAME_0=="Libya"         | ///
		NAME_0=="Madagascar"    | NAME_0=="Malawi"      | NAME_0=="Mali"          | NAME_0=="Mauritania"    | ///
		NAME_0=="Mauritius"     | NAME_0=="Morocco"     | NAME_0=="Mozambique"    | NAME_0=="Namibia"       | ///
		NAME_0=="Niger"         | NAME_0=="Nigeria"     | NAME_0=="Rwanda"        | NAME_0=="São Tomé and Príncipe" | ///
		NAME_0=="Senegal"       | NAME_0=="Seychelles"  | NAME_0=="Sierra Leone"  | NAME_0=="Somalia"       | ///
		NAME_0=="South Africa"  | NAME_0=="South Sudan" | NAME_0=="Sudan"         | NAME_0=="Swaziland"     | ///
		NAME_0=="Tanzania"      | NAME_0=="Togo"        | NAME_0=="Tunisia"       | NAME_0=="Uganda"        | ///
		NAME_0=="Zambia"        | NAME_0=="Zimbabwe"    | NAME_0 == "Western Sahara"
	merge 1:1 NAME_0 using "../../Data/02. Peacekeeping/UNPO8_financial_resources_base_map.dta"
	drop if _merge == 2
	drop _merge

	*grmap, activate
	grmap max_civilianexp, ///
		clbreaks(0(100)1000) clnumber(10) clm(custom) ndf(none) ndlabel("No PKOs")       ///
		fc(pink%10 pink%20 pink%30 pink%40 pink%50 pink%60 pink%70 pink%80 pink%90 pink) ///
		legtitle("Civilian expenditures" "(# of countries)") legend(size(medsmall)) legstyle(2) legcount
	*graph export "../Figures/Fig04_Map_of_civilian_expenditures_in_Africa.pdf", as(pdf) replace
	/*I make a .png file by screenshot because .pdf file is too heavy to show.*/

	cd "../.."




********************************************************************************
/*Descriptive statistics*/
	use "Data/Dataset_monthly.dta", clear

/*Table 1*/
	foreach var of varlist pko ln_srf_civilianexp ln_srf_militaryexp ln_srf_policeexp spm {
		gen l_`var' = l.`var'
	}
	estpost sum ///
		bd_best_ged bd_govt_ged bd_rebel_ged                                      ///
		l_pko l_ln_srf_civilianexp l_ln_srf_militaryexp l_ln_srf_policeexp        ///
		pa bd_total_osv_govt_binary bd_total_osv_rebel_binary bd_total_nsv_binary ///
		ln_oda ln_milper ln_gdp ln_pop v2x_polyarchy l_spm ropko stpko time
	esttab, /*tex*/ ///
		cells("count(fmt(%9.0gc) label(N)) mean(fmt(%12.3fc) label(Mean)) sd(fmt(%12.3fc) label(Std. Dev.)) min(fmt(a5) label(Min.)) max(fmt(a5) label(Max.))") ///
		nogap label nonumbers nomtitles varwidth(40) modelwidth(11) coll(,lhs("Variables")) noobs title("Descriptive statistics") ///
		coeflabels(bd_best_ged "Total Battle-Related Deaths" bd_govt_ged "Government Battle-Related Deaths"                       ///
		           bd_rebel_ged "Rebel Battle-Related Deaths" l_pko "UN PKO\(_{t-1}\)" l_ln_srf_civilianexp "Civilian Expenditures\(_{t-1}\)\(^{a}\)" ///
				   l_ln_srf_militaryexp "Military Expenditures\(_{t-1}\)\(^{a}\)" l_ln_srf_policeexp "Police Expenditures\(_{t-1}\)\(^{a}\)" ///
				   pa "Peace Agreement" bd_total_osv_govt_binary "Government OSV" bd_total_osv_rebel_binary "Rebel OSV"             ///
				   bd_total_nsv_binary "Non-State Violence" ln_oda "log(Official Development Assistance)\(^{b}\)"                   ///
				   ln_milper "log(Army Size)\(^{b}\)" ln_gdp "log(Gross Domestic Product)\(^{b}\)" ln_pop "log(Population)\(^{b}\)" ///
				   v2x_polyarchy "Democracy\(^{b}\)" l_spm "Special Political Mission\(_{t-1}\)" ropko "Regional PKO\(^{b}\)" stpko "State PKO\(^{b}\)" time "t") ///
		addnotes("Note: The unit of analysis is a conflict-month. OSV = One-Sided Violence."                                 ///
				 "\(^{a}\)Measured in thousand USD, on a daily average, per logarithmic squared kilometer; used in UN PKOs." ///
				 "\(^{b}\)Measured on a yearly basis.")


/*Fn. 8*/
	/*Active year*/
	sort conflict_id year month
	by conflict_id year: egen bd_best_ged_yr = sum(bd_best_ged)
	gen activeyr = 0
	replace activeyr = 1 if bd_best_ged_yr >= 25

	/*Month with at least some PKO expenditures*/
	gen withexp = 0 if ln_srf_militaryexp != . & ln_srf_policeexp != . & ln_srf_civilianexp != .
	recode withexp (0=1) if ln_srf_militaryexp != 0 | ln_srf_policeexp != 0 | ln_srf_civilianexp != 0

	/*PKO existence in in/active years*/
	tab activeyr withexp, col





********************************************************************************
/*Regression analysis*/
	use "Data/Dataset_monthly.dta", clear
	global pkexps   = "l.ln_srf_civilianexp l.ln_srf_militaryexp l.ln_srf_policeexp"
	global controls = "pa bd_total_osv_govt_binary bd_total_osv_rebel_binary bd_total_nsv_binary ln_oda ln_milper ln_gdp ln_pop v2x_polyarchy l.spm ropko stpko"

/*Table 2*/
	/*(1) DV = all BRDs*/
	xtreg bd_best_ged $pkexps $controls i.ym c.time#i.conflict_id, fe cluster(conflict_id)
	estadd local War_FEs  "YES"
	estadd local YM_FEs   "YES"
	estadd local War_FEsT "YES"
	est sto m1

	/*(2) DV = BRDs on the government side*/
	xtreg bd_govt_ged $pkexps $controls i.ym c.time#i.conflict_id, fe cluster(conflict_id)
	estadd local War_FEs  "YES"
	estadd local YM_FEs   "YES"
	estadd local War_FEsT "YES"
	est sto m2

	/*(3) DV = BRDs on the rebel side*/
	xtreg bd_rebel_ged $pkexps $controls i.ym c.time#i.conflict_id, fe cluster(conflict_id)
	estadd local War_FEs  "YES"
	estadd local YM_FEs   "YES"
	estadd local War_FEsT "YES"
	est sto m3

	esttab m1 m2 m3, /*tex*/ ///
		nogap compress unstack noomitted drop(*.ym *.conflict_id#c.time _cons) b(3) se(3)         ///
		stat(War_FEs YM_FEs War_FEsT N r2_w, label("War FEs" "Year-month FEs" "War FEs * t" "N" "R-squared (within)") ///
		fmt(%3s %3s %3s %9.0gc %12.3fc)) star(+ 0.10 * 0.05 ** 0.01 *** 0.001)                    ///
		title("Fixed effects OLS regression results on battle-related deaths in intrastate conflicts, 1989-2018") ///
		/*numbers("Model ")*/ varwidth(35) modelwidth(10) mlabels("Total" "Government" "Rebel")   ///
		coeflabels(L.ln_srf_civilianexp "Civilian Expenditures\(_{t-1}\)" L.ln_srf_militaryexp "Military Expenditures\(_{t-1}\)" L.ln_srf_policeexp "Police Expenditures\(_{t-1}\)" ///
				   pa "Peace Agreement" bd_total_osv_govt_binary "Government OSV"                 ///
				   bd_total_osv_rebel_binary "Rebel OSV" bd_total_nsv_binary "Non-State Violence" ///
				   ln_oda "log(ODA)" ln_milper "log(Army Size)" ln_gdp "log(GDP)" ln_pop "log(Population)" ///
				   v2x_polyarchy "Democracy" L.spm "SPM\(_{t-1}\)" ropko "Regional PKO" stpko "State PKO") ///
		note("Robust standard errors clustered at conflict in parentheses.")





********************************************************************************
/*Interpretations (using Model 2)
  Note: Predicting the number of deaths may not be appropriate when using these linear models because of many zeros and negative prediction.
        Thus, changes in the number of battle-related deaths are used for substantive interpretation.*/
	xtreg bd_govt_ged $pkexps $controls i.ym c.time#i.conflict_id, fe cluster(conflict_id)
	matrix define model2 = r(table)
	scalar coef = model2[1,1]

	/*# of zeroes = distribution of battle-related deaths*/
	count if e(sample)
	count if e(sample) & bd_govt_ged == 0
	dis 38632 / 44441                               /*86.93% in the sample are 0s.*/
	codebook bd_govt_ged if e(sample)               /*90th percentile is 3.*/
	sum bd_govt_ged if e(sample) & bd_govt_ged >= 1 /*Average # of deaths is 41 in the sample with nonzero deaths.*/
	*sum bd_govt_ged if bd_govt_ged >= 1             /*Average = 41 (same)*/

	/*1 standard deviation shift*/
	sum L.ln_srf_civilianexp if e(sample)
	scalar sd = r(sd)        /*changes in x: +6,485 USD on a daily average, per log(squared kilometer).*/
	dis coef * sd            /*changed in y: -8.91 battle-related deaths on the government side.*/

	/*Another measure for a 1 standard deviation shift*/
	gen withexp = 0 if ln_srf_militaryexp != . & ln_srf_policeexp != . & ln_srf_civilianexp != .
	recode withexp (0=1) if ln_srf_militaryexp != 0 | ln_srf_policeexp != 0 | ln_srf_civilianexp != 0
	sum L.ln_srf_civilianexp if e(sample)                /*In-sample                         :  1,177 USD on average*/
	sum L.ln_srf_civilianexp if e(sample) & withexp == 1 /*In-sample with UN PKO exp.        : 18,333 USD*/
	sum L.ln_srf_civilianexp if withexp == 1             /*In-/Out-of-sample with UN PKO exp.: 19,478 USD*/
	scalar sdpko = r(sd)     /*changes in x: +18,502 USD on a daily average, per log(squared kilometer).*/
	dis coef * sdpko         /*changes in y: -25.42 battle-related deaths.*/


	/*How large is 6,485 USD (on a daily average, per logged squared kilometer)? (Note: I focus on Africa.)*/
	sum srf if region == 4 & withexp == 1               /*min: Rwanda; max: DRC*/
	dis 6485 * ln(26340)                                /*6,485 USD per log(km^2) is equivalent to 66,010 USD in Rwanda.*/
	sum L.civilianexp if location == "Rwanda"           /*Actual max civilian expenditure for Rwanda is 83,665 USD.*/
	*sum L.civilianexp if location == "Rwanda" & withexp == 1 /*39,531 USD on average in Rwanda.*/
	dis (6485 * ln(26340)) / 83665                      /*That is, 6,485 USD per log(km^2) is equivalent to 78.90% of civilian expenditure for Rwanda.*/
	dis 6485 * ln(2344860)                              /*6,485 USD per log(km^2) is equivalent to 95,120 USD in DRC.*/
	sum L.civilianexp if location == "DR Congo (Zaire)" /*Actual max civilian expenditure for DRC is 980,617 USD.*/
	*sum L.civilianexp if location == "DR Congo (Zaire)" & withexp == 1 /*592,902 USD on average in DRC.*/
	dis (6485 * ln(2344860)) / 980617                   /*That is, 6,485 USD per log(km^2) is equivalent to 9.70% of civilian expenditure for DRC.*/

	dis 18502 * ln(26340)                               /*18,502 USD per log(km^2) is equivalent to 188,329 USD in Rwanda.*/
	dis (18502 * ln(26340)) / 83665                     /*18,502 USD per log(km^2) is 2.25 times as large as actual civilian expenditure for Rwanda.*/
	dis 18502 * ln(2344860)                             /*18,502 USD per log(km^2) is equivalent to 271,382 USD in DRC.*/
	dis (18502 * ln(2344860)) / 980617                  /*18,502 USD per log(km^2) is equivalent to 27.67% of civilian expenditure for DRC.*/


	/*How many peacekeepers are equivalent to 6,485 USD (or 18,502 USD) per log(km^2)?*/
	gen civilianexp_pc = (civilianexp * 1000) / civilian
	gen ln_civilianexp_pc = ln(civilianexp_pc)
	*hist ln_civilianexp_pc
	codebook *civilianexp_pc /*median: 189 USD per day*/

	codebook *civilianexp_pc if location == "Rwanda"
	*twoway line civilian ym if conflict_id == 374 ||  line civilianexp ym if conflict_id == 374 || line civilianexp_pc ym if conflict_id == 374, yaxis(2)
	codebook *civilianexp_pc if location == "DR Congo (Zaire)"
	*twoway line civilian ym if conflict_id == 283 ||  line civilianexp ym if conflict_id == 283 || line civilianexp_pc ym if conflict_id == 283, yaxis(2)

	*codebook civilian if civilian != 0 & location == "Rwanda"
	dis 66010 / 189  /*349 civilian pks*/
	dis 349 / 8      /*43*/
	*dis 188329 / 189
	*dis 996 / 25
	*codebook civilian if civilian != 0 & location == "DR Congo (Zaire)"
	dis 95120 / 189 /*503 civilian pks*/
	dis 503 / 8     /*62*/
	*dis 271382 / 189
	*dis 1435 / 25





********************************************************************************
/*Robustness Checks -> See 03_Appendix.do*/




